

**Excel & Business Math**  
**Video/Class Project #33**

**VLOOKUP Function for Incentive Pay: Commissions and Piecework**

**Topics**

1) Incentive Pay.....	2
2) Straight Piecework Example .....	3
3) Variable Piecework Example Making the Calculations Long-Hand .....	3
4) Why Excel is so Helpful AND why we must be REALLY Knowledgeable with Excel.....	4
5) Variable Piecework Example Making the Calculations with the “Lookup Table Method” .....	4
6) Creating the Lookup Table from Scratch .....	7
7) What is VLOOKUP Function? .....	9
8) Variable Piecework Example Making the Calculations with the “Lookup Table Method & VLOOKUP Function using Approximate Match” .....	10
9) Variable Piecework & VLOOKUP & Approximate Match Lookup for Entire Payroll Table: .....	11
10) “Mash” all 3 VLOOKUP Functions into Single Formula using Clipboard:.....	12
11) VLOOKUP Function Arguments (Full Details): .....	13
12) Lookup Tables are Everywhere.....	14
13) Straight Commission Rates .....	15
14) VLOOKUP & Exact Match Lookup for Straight Commission Rates When each Employee has Different Rate .....	15
15) VLOOKUP & Approximate Match Lookup for Variable Commission Rates .....	16
16) VLOOKUP & Approximate Match Lookup for Commission Rate Based on Amount of Sales Made by the Employee.....	17

## 1) Incentive Pay

- 1) **Incentive Pay Rates:** This means you get paid based on your performance, such as "How Many Items You Made" or "How Much You Sold"
- 2) Incentive pay: the more you make or sell, the more you are paid.
- 3) Examples:
  - i. Boomerang Company Pays employees in various ways:
    1. Employees making boomerangs may have a contract that says:
      - i. **Straight Piecework Rate:** Manufacturer gets paid \$1.35 per one Bellen Boomerang made
      - ii. **Variable Piecework Rate:** Manufacturer gets paid \$1.05 for 1 - 75 Boomerangs and \$1.35 for 76 - 120 Boomerangs and \$1.75 for 121 or more Boomerangs
    2. Employees making sales calls may have a contract that says:
      - i. **Straight Commission Rate:** Sales person gets a base monthly salary of \$2,000 plus 1.5% of the total sales they made for the month
      - ii. **Variable Commission Rate:** Sales person gets a base monthly salary of \$1,500 plus 1% for sales from \$0 – \$9,999.99, 1.5% for sales from \$10,000 – \$19,999.99 and 3% for sales of \$20,000 or more.
      - iii. **Commission Rate Based on Sales Amount:** Sales person gets a base monthly salary of \$2,000 plus a percentage of the total sales they made for the month based on the amount of sales they make (table of rates is in below picture).
  - ii. Picture of Excel Sheet in Video:

### Incentive Pay Rates

This means you get paid based on your performance, such as "How Many Items You Made" or "How Much You Sold".

**Incentive pay:** the more you make or sell, the more you are paid.

### Examples:



Bellen Boomerang:



### **Boomerang Company Pays employees in various ways:**

#### **Employees making boomerangs may have a contract that says:**

**Straight Piecework Rate:** Manufacturer gets paid \$1.35 per one Bellen Boomerang made.

**Variable Piecework Rate:** Manufacturer gets paid a different rate based on quantity made:

1 - 75 ==>	\$1.05 each.
76 - 120 ==>	\$1.35 each.
121 or more ==>	\$1.75 each.

#### **Employees making sales calls may have a contract that says:**

**Straight Commission Rate:** Sales person gets a base monthly salary of \$3,000 plus 1.5% of the total sales they made for the month.

**Variable Commission Rate:** Sales person gets a base monthly salary of \$3,000 plus 1% for sales from \$0 – \$9,999.99, 1.5% for sales from \$10,000 – \$19,999.99 and 3% for sales of \$20,000 or more:

0.00 - \$9,999.99 ==>>	1.0%
\$10,000.00 - \$19,999.99 ==>>	1.5%
\$20,000.00 or more ==>>	3.0%

**Commission Rate Based on Sales Amount:** Sales person gets a base monthly salary of \$2,000 plus a percentage of the total sales they made for the month based on the amount of sales they made determined by this table:

0.00 - \$4,999.99 ==>>	0.50%
\$5,000.00 - \$19,999.99 ==>>	2.00%
\$20,000.00 - \$29,999.99 ==>>	2.50%
\$30,000.00 - \$39,999.99 ==>>	3.75%
\$40,000 or more ==>>	5.25%

## 2) Straight Piecework Example

	A	B	C	D	E	F	G	H	
1	<b>Ex 1</b>	<b>Goal:</b> Calculate Gross Pay for Straight Piecework.							
2		<b>Contact reads:</b> Pay \$1.35 per one Bellen Boomerang made.							
3		Pay per 1 Bellen Boomerang	\$1.35						
4		# of Bellens made	127						
5		Gross Pay	\$171.45		=C4*C3				
6									
7	<b>Ex 2</b>	<b>Goal:</b> Calculate Gross Pay for Straight Piecework.							
8		<b>Contact reads:</b> Pay straight rate: Small Lamps@ \$1.01, Shades@ \$0.275, Large Lamps@ \$1.275.							
9		<b>Item Made</b>	<b>Number Made</b>	<b>Pay per 1 Made</b>	<b>Total</b>				
10		Small Lamps	39	\$1.010	39.39			=ROUND(D10*C10,2)	
11		Shades	112	\$0.275	30.8				
12		Large Lamps	21	\$1.275	26.78				
13				Gross Pay	96.97			=SUM(E10:E12)	

## 3) Variable Piecework Example Making the Calculations Long-Hand

1) Why we should NOT use Long-Hand Method:

- i. Takes too long.
- ii. Risks making mistakes when you are making many calculations to get to a final answer.

	A	B	C	D	E	F	G	H	I
1	<b>Ex 3</b>	<b>Goal:</b> Calculate Gross Pay for Variable Piecework Long-Hand Method.							
2		Employee made 137 boomerangs.							
3		<b>Contact reads:</b> Pay Rate:							
4		\$1.05 for 1 - 75 Boomerangs.							
5		\$1.35 for 76 - 120 Boomerangs.							
6		\$1.75 for 121 or more Boomerangs.							
7									
8		<b>Long Method:</b>							
9		Boomerangs Made	137						
10		Category #1 Upper Limit	75						
11		Paid per Unit for #1 Category	1.05						
12		Gross Pay #1	\$78.75		=C11*C10				
13		Remains after #1	62		=C9-C10				
14		Category #2 Upper Limit	120						
15		Number possible in Category #2	45		=C14-C10				
16		Paid per Unit for #2 Category	1.35						
17		Gross Pay #2	\$60.75		=C16*C15				
18		Remains after #2	17		=C13-C15				
19		Paid per Unit for #3 Category	1.75						
20		Gross Pay #3	\$29.75		=C19*C18				
21		Total Gross Pay	\$169.25		=C20+C17+C12				
22									
23									

category #1 0 - 75 = \$1.05 per 1 Boom  
category #2 76 - 120 = \$1.35 per 1 Boom  
category #3 121 - ?? = \$1.75 per 1 Boom

Employee:  
made Booms = 137 <sup>check: bigger?</sup>

\* category #1 Gross Pay = 75 \* 1.05 = \$78.75

# Booms Remain after category #1 = 137 - 75 = 62

# Booms for category #2 = 120 - 75 = 45 <sup>check bigger?</sup>

\* category #2 Gross Pay = 45 \* 1.35 = \$60.75

# Booms Remain after category #2 = 62 - 45 = 17

\* category #3 Gross Pay = 17 \* 1.75 = \$29.75

Total Gross Pay = 78.75 + 60.75 + 29.75 =  
\$169.25

#### 4) Why Excel is so Helpful AND why we must be REALLY Knowledgeable with Excel.

- 1) In example #3, the business has an established method for paying the employee.
  - i. But the Accounting Program that they use, QuickBooks, does not have a built-in feature to make this payroll calculation.
  - ii. We would not want to have to make this calculation Long-Hand, either on paper or in Excel, every time an employee worked to make boomerangs.
    1. Why we should NOT use Long-Hand Method:
      - i. Takes too long.
      - ii. Risks making mistakes when you are making many calculations to get to a final answer.
  - iii. Luckily, we can automate the calculations using two methods:
    1. Lookup Table Method.
    2. Lookup Table Method & VLOOKUP Function.
- 2) Why Excel is so Helpful AND why we must be REALLY Knowledgeable with Excel:
  - i. Excel is helpful because with Excel we can take all the calculations in example #3 and make them more automated, as long as we are knowledgeable enough with Excel to create the correct Lookup Table and then use the VLOOKUP Function.

#### 5) Variable Piecework Example Making the Calculations with the “Lookup Table Method”

- 1) Steps for calculating Gross Pay using Lookup Table Method:
  1. Determine Total Number of Items Made
  2. Using the total number of items made, find correct category (row) in the Lookup Table
  3. Get the amount for Earnings Made Through Previous Category
  4. Get Number Made Through Previous Category
  5. Get Pay per Unit
  6. Make the calculation: Earnings Made Through Previous Category + (Total Number of Items Made - Number Made Through Previous Category) \* Pay per Unit

	A	B	C	D	E	F	G	H
1	<b>Example 4:</b>	<b>Goal:</b> Calculate Gross Pay for Variable Piecework using Lookup Table Method.						
2		Employee made 137 boomerangs.						
3		Contract shows the Lookup Table Below with the Pay Data for Each Category						
4								
5		<b>Lookup Table below with the pay data for each category</b>						
6		<b>Columns in Lookup Table:</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>	
7			Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category	
8		0- 75	0	75	1.05	0	0	
9		76- 120	76	120	1.35	75	78.75	
10	Using 137 pick category →	121- more	121	more	1.75	120	139.5	
11								
12		<b>Boomerangs Made</b>	137					
13		<b>Gross Pay</b>	169.25				=G10+(C12-F10)*E10 = 139.5 + (137 - 120) * 1.75 = 169.25	
14								
15		<b>Steps for Calculating Gross Pay using Lookup Table Method</b>						
16		1) Determine total number of items made: 137						
17		2) Using the total number of items made, find correct category (row) in the Lookup Table: 121- more						
18		3) From Column 5, get the amount for Earnings Made Through Previous Category: 139.5						
19		4) From Column 4 get Number Made Through Previous Category: 120						
20		5) From Column 3 get Pay per Unit: 1.75						
21		6) Make the calculation: 139.5 + (137 - 120) * 1.75 = 169.25						

**Example 5:** **Goal:** Calculate Gross Pay for Variable Piecework using Lookup Table Method.  
Employee made 119 boomerangs.  
Contract shows the Lookup Table Below with the Pay Data for Each Category

**Lookup Table below with the pay data for each category**

Columns in Lookup Table:    **1**        **2**        **3**        **4**        **5**

Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
0- 75	0	75	1.05	0	0
76- 120	76	120	1.35	75	78.75
121- more	121	more	1.75	120	139.5

Using 119 pick category →

Boomerangs Made	119
Gross Pay	138.15

$$=G31+(C34-F31)*E31 = 78.75 + (119 - 75) * 1.35 = 138.15$$

**Steps for Calculating Gross Pay using Lookup Table Method**

- 1) Determine total number of items made: 119
- 2) Using the total number of items made, find correct category (row) in the Lookup Table: 76- 120
- 3) From Column 5, get the amount for Earnings Made Through Previous Category: 78.75
- 4) From Column 4 get Number Made Through Previous Category: 75
- 5) From Column 3 get Pay per Unit: 1.35
- 6) Make the calculation:  $78.75 + (119 - 75) * 1.35 = 138.15$

**Example 6:** **Goal:** Calculate Gross Pay for Variable Piecework using Lookup Table Method.  
Employee made 74 boomerangs.  
Contract shows the Lookup Table Below with the Pay Data for Each Category

**Lookup Table below with the pay data for each category**

Columns in Lookup Table:    **1**        **2**        **3**        **4**        **5**

Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
0- 75	0	75	1.05	0	0
76- 120	76	120	1.35	75	78.75
121- more	121	more	1.75	120	139.5

Using 74 pick category →

Boomerangs Made	74
Gross Pay	77.7

$$=G52+(C56-F52)*E52 = 0 + (74 - 0) * 1.05 = 77.7$$

**Steps for Calculating Gross Pay using Lookup Table Method**

- 1) Determine total number of items made: 74
- 2) Using the total number of items made, find correct category (row) in the Lookup Table: 0- 75
- 3) From Column 5, get the amount for Earnings Made Through Previous Category: 0
- 4) From Column 4 get Number Made Through Previous Category: 0
- 5) From Column 3 get Pay per Unit: 1.05
- 6) Make the calculation:  $0 + (74 - 0) * 1.05 = 77.7$

	A	B	C	D	E	F	G	H
--	---	---	---	---	---	---	---	---

67 **Example 7:** **Goal:** Calculate Gross Pay for Variable Piecework using Lookup Table Method.  
 68 Employee made 137 boomerangs.  
 69 Contract shows the Lookup Table Below with the Pay Data for Each Category

**Lookup Table below with the pay data for each category**

72 **Columns in Lookup Table:**

	1	2	3	4	5
Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
0- 50	0	50	\$1.00	0	0
51- 75	51	75	\$1.10	50	\$50.00
76- 110	76	110	\$1.25	75	\$77.50
111- 130	111	130	\$1.50	110	\$121.25
131- 150	131	150	\$1.75	130	\$151.25
151- more	151	more	\$2.00	150	\$186.25

78 Using 137 pick category →

Boomerangs Made	137
Gross Pay	\$163.50

=G78+(C81-F78)\*E78 = 151.25 + (137 - 130) \* 1.75 = 163.5

**Steps for Calculating Gross Pay using Lookup Table Method**

- 1) Determine total number of items made: 137
- 2) Using the total number of items made, find correct category (row) in the Lookup Table: 131- 150
- 3) From Column 5, get the amount for Earnings Made Through Previous Category: 151.25
- 4) From Column 4 get Number Made Through Previous Category: 130
- 5) From Column 3 get Pay per Unit: 1.75
- 6) Make the calculation: 151.25 + (137 - 130) \* 1.75 = 163.5

## 6) Creating the Lookup Table from Scratch

- Most of the times we are not given the Lookup Table and so we have to make it. Sometimes it is difficult to make it, but it is always worth it because then you can use it over and over.
- Here is a description of the categories for each pay rate:
  - Contact reads: Pay Rate:
    - \$1.00 for 0- 50 Boomerangs
    - \$1.10 for 51- 75 Boomerangs
    - \$1.25 for 76- 110 Boomerangs
    - \$1.50 for 111- 130 Boomerangs
    - \$1.75 for 131- 150 Boomerangs
    - \$2.00 for 151- More Boomerangs
- Below is a picture of the formulas needed to create the Lookup Table:

	A	B	C	D	E	F	G	H
10		Now you have to create Lookup Table From Scratch, so you can use it over and over.						
11		Now you have to create Lookup Table From Scratch, so you can use it over and over.						
12								
13		Formula in B18:	Formula in C19:			Formula in F19:	Formula in G19:	
14		=C18&"- "&D18	=D18+1			=D18	=(F19-F18)*E18+G18	# Units Made in Previous Category * Pay per Unit in Previous Category + Earnings Made Through Previous Category
15								
16		<b>Columns in Lookup Table:</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>	
17		Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category	
18		0- 50	0	50	\$1.00			
19		51- 75	51	75	\$1.10	50	\$50.00	
20		76- 110	76	110	\$1.25	75	\$77.50	
21		111- 130	111	130	\$1.50	110	\$121.25	
22		131- 150	131	150	\$1.75	130	\$151.25	
23		151- More	151	More	\$2.00	150	\$186.25	
24								
25		Boomerangs Made	137					
26		Gross Pay	\$163.50		=G22+(C25-F22)*E22			
27		Check:	\$163.50		=D18*E18+(D19-D18)*E19+(D20-D19)*E20+(D21-D20)*E21+(C25-D21)*E22			

4. Below is close up for the “Earnings Made Through Previous Category” Formula:

	C	D	E	F	G
16	1	2	3	4	5
	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
18	0	50	\$1.00		
19	51	75	\$1.10	50	\$50.00
20	76	110	\$1.25	75	$=(F20-F19)*E19+G19$
21	111	130	\$1.50	110	\$121.25

**E19**  
Pay per Unit in Previous Category

**G19**  
Earnings Made Through Previous Category

**(F20-F19)**  
# Units Made in Previous Category

“Earnings Made Through Previous Category” Formula



5. Once we create our Lookup Table, we can use it for new employees, like seen here for the employees, Abdi, Gigi and Kenya:

	A	B	C	D	E	F	G
16		<b>Columns in Lookup Table:</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>
17		Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
18		0- 50	0	50	\$1.00		
19		51- 75	51	75	\$1.10	50	\$50.00
20		76- 110	76	110	\$1.25	75	\$77.50
21		111- 130	111	130	\$1.50	110	\$121.25
22		131- 150	131	150	\$1.75	130	\$151.25
23		151- More	151	More	\$2.00	150	\$186.25
28							
29	<b>Ex 9</b>	Now we can use table over and over:					
30							
31		Employee	Number Made	Gross Pay			
32		Abdi Smitty	116	\$130.25		=G21+(C32-F21)*E21	
33		Gigi Thmpson	72	\$74.20		=G19+(C33-F19)*E19	
34		Kenya Panther	152	\$190.25		=G23+(C34-F23)*E23	

## 7) What is VLOOKUP Function?

- 1) What does VLOOKUP Function do?

VLOOKUP tries to find a match of an item in the first column of the Lookup Table and then retrieves (goes and gets) something from one of the other columns in the table and bring it back to the cell or formula.

- 2) In VLOOKUP the V means Vertical because the categories in the Lookup Table are oriented Vertically.
- 3) Example: VLOOKUP can find a match for the Number of Boomerangs Made (137) in the sorted first column of the Lookup Table and retrieve the correct “Earnings Made Through Previous Category” (\$151.25) from the 5th column and bring it back to the cell C13, like in this picture:

	B	C	D	E	F	G
3	<b>Columns in Lookup Table:</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>
4	Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
5	0- 50	0	50	\$1.00		
6	51- 75	51	75	\$1.10	50	\$50.00
7	76- 110	76	110	\$1.25	75	\$77.50
8	111- 130	111	130	\$1.50	110	\$121.25
9	131- 150	131	150	\$1.75	130	\$151.25
10	151- More	151	More	\$2.00	150	\$186.25
11						
12	Boomerangs Made	137				
13	Earnings Made Through Previous Category	=VLOOKUP(C12,C5:G10,5)		=VLOOKUP(C12,C5:G10,5)		
14	Number Made Through Previous Category	130		=VLOOKUP(C12,C5:G10,4)		
15	Pay per Unit	\$1.75		=VLOOKUP(C12,C5:G10,3)		
16	Gross Pay	\$163.50		=C13+(C12-C14)*C15		

### 8) Variable Piecework Example Making the Calculations with the “Lookup Table Method & VLOOKUP Function using Approximate Match”

	A	B	C	D	E	F	G	H
1	<b>Ex 10</b>	<b>Goal:</b> Use VLOOKUP Function (with Approximate Match Lookup) & Lookup Table to automate calculation for Variable Piecework Gross Pay.						
2								
3		<b>Columns in Lookup Table:</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>	
4		Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category	
5		0- 50	0	50	\$1.00			
6		51- 75	51	75	\$1.10	50	\$50.00	
7		76- 110	76	110	\$1.25	75	\$77.50	
8		111- 130	111	130	\$1.50	110	\$121.25	
9		131- 150	131	150	\$1.75	130	\$151.25	
10		151- More	151	More	\$2.00	150	\$186.25	
11								
12		Boomerangs Made	137					
13		Earnings Made Through Previous Category	\$151.25		=VLOOKUP(C12,C5:G10,5)			
14		Number Made Through Previous Category	130		=VLOOKUP(C12,C5:G10,4)			
15		Pay per Unit	\$1.75		=VLOOKUP(C12,C5:G10,3)			
16		Gross Pay	\$163.50		=C13+(C12-C14)*C15			
17								
18		<b>VLOOKUP Function Notes:</b>						
19		=VLOOKUP(lookup_value,table_array,col_index_num,[range_lookup])						
20		lookup_value: Item you are trying to find a match for in in first column of lookup table (find first bigger one and jump back a row)						
21		table_array: Lookup Table (1st column must have Smallest Number for each Category AND 1st column must be sorted smallest to biggest)						
22		col_index_num: Which Columns has the Number you want to go and get and bring back to cell? 1, 2, 3, 4, 5?						
23		[range_lookup]: Approximate Match (TRUE or omitted) allows VLOOKUP to find correct category when 1st column is sorted smallest to biggest.						
24								

### 9) Variable Piecework & VLOOKUP & Approximate Match Lookup for Entire Payroll Table:

	A	B	C	D	E	F	G	H
1	<b>Ex 11</b>	<b>Goal:</b> Use Excel's Power to do many Variable Piecework Calculations all at once!!						
2		Use VLOOKUP Function (with Approximate Match Lookup) & Lookup Table to automate calculation for Variable Piecework Gross Pay.						
3								
4				Formula is D8:	Formula is E8:	Formula is F8:	Formula is G8:	
5				=VLOOKUP(C8,\$C\$19:\$G\$24,5)	=VLOOKUP(C8,\$C\$19:\$G\$24,4)	=VLOOKUP(C8,\$C\$19:\$G\$24,3)	=D8+(C8-E8)*F8	
6								
7		<b>Employee</b>	<b>Number Made</b>	<b>Earnings Made Through Previous Category</b>	<b>Number Made Through Previous Category</b>	<b>Pay per Unit</b>	<b>Gross Pay</b>	
8		Abdi Smitty	116	121.25	110	1.5	\$130.25	
9		Gigi Thompson	72	50	50	1.1	\$74.20	
10		Kenya Panther	152	186.25	150	2	\$190.25	
11		Dylan Minger	45	0	0	1	\$45.00	
12		Chin Yu	140	151.25	130	1.75	\$168.75	
13		Billy Smith	139	151.25	130	1.75	\$167.00	
14		Lin Pham	119	121.25	110	1.5	\$134.75	
15		Shelia Downings	91	77.5	75	1.25	\$97.50	
16								
17		<b>Columns in Lookup Table:</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>	
18		<b>Category</b>	<b>Units Lower Limit</b>	<b>Units Upper Limit</b>	<b>Pay per Unit</b>	<b>Number Made Through Previous Category</b>	<b>Earnings Made Through Previous Category</b>	
19		0- 50	0	50	\$1.00			
20		51- 75	51	75	\$1.10	50	\$50.00	
21		76- 110	76	110	\$1.25	75	\$77.50	
22		111- 130	111	130	\$1.50	110	\$121.25	
23		131- 150	131	150	\$1.75	130	\$151.25	
24		151- More	151	More	\$2.00	150	\$186.25	
25								

## 10) "Mash" all 3 VLOOKUP Functions into Single Formula using Clipboard:

1. Rather than use three columns with an individual VLOOKUP in each column, we can create a single column formula.
2. To make the formula easy to create, you can open the Clipboard, copy each VLOOKUP Function from the cell while the cell is in Edit Mode and collect the three VLOOKUP Function Formula Elements in the Clipboard.
3. You can paste the individual VLOOKUP Functions into a single cell formula when the single cell is in edit mode by clicking on the VLOOKUP Formula Element in the Clipboard.
4. Just as the Formula in G8 is =D8+(C8-E8)\*F8, you can replace the cell references with the correct VLOOKUP Formula Element to get:
  - $V5 + (Units - V4) * V3$
  - or
  - =VLOOKUP(C8,\$C\$19:\$G\$24,5)+(C8-VLOOKUP(C8,\$C\$19:\$G\$24,4))\*VLOOKUP(C8,\$C\$19:\$G\$24,3)

Open Clipboard by clicking the Clipboard Dialog Launcher

Clipboard

Click an Item to Paste:

- =VLOOKUP(C8,\$C\$19:\$G\$24,3)
- =VLOOKUP(C8,\$C\$19:\$G\$24,4)
- =VLOOKUP(C8,\$C\$19:\$G\$24,5)

Goal: Use Clipboard to collect each VLOOKUP Function (that you copy in edit mode) and then paste them all into a single formula.  
Use VLOOKUP Function (with Approximate Match Lookup) & Lookup Table to automate calculation for Variable Piecework Gross Pay.

Employee	Number Made	Earnings Made Through Previous Category	Number Made Through Previous Category	Pay per Unit	Gross Pay	Gross Pay (all together)	V5 + (Units - V4)*V3
Abdi Smitty	116	121.25	110	1.5	\$130.25		=VLOOKUP(C8,\$C\$19:\$G\$24,5)+(C8-VLOOKUP(C8,\$C\$19:\$G\$24,4))*VLOOKUP(C8,\$C\$19:\$G\$24,3)
Gigi Thompson	72	50	50	1.1	\$74.20		
Kenya Panther	152	186.25	150	2	\$190.25	\$190.25	
Dylan Minger	45	0	0	1	\$45.00	\$45.00	
Chin Yu	140	151.25	130	1.75	\$168.75	\$168.75	
Billy Smith	139	151.25	130	1.75	\$167.00	\$167.00	
Lin Pham	119	121.25	110	1.5	\$134.75	\$134.75	
Shelia Downings	91	77.5	75	1.25	\$97.50	\$97.50	

Columns in Lookup Table:

Category	Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous Category	Earnings Made Through Previous Category
0- 50	0	50	\$1.00		
51- 75	51	75	\$1.10	50	\$50.00
76- 110	76	110	\$1.25	75	\$77.50
111- 130	111	130	\$1.50	110	\$121.25
131- 150	131	150	\$1.75	130	\$151.25
151- More	151	More	\$2.00	150	\$186.25

Formula is I8: =VLOOKUP(C8,\$C\$19:\$G\$24,5)+(C8-VLOOKUP(C8,\$C\$19:\$G\$24,4))\*VLOOKUP(C8,\$C\$19:\$G\$24,3)


## 11)VLOOKUP Function Arguments (Full Details):

=VLOOKUP( lookup\_value , table\_array , col\_index\_num , [range\_lookup] )

- i. **lookup\_value** = Item that you are trying to find in first column of lookup table.
- ii. **table\_array** = Vertical table = Lookup table. First Column contains items you want to “match” with the lookup\_value.
- iii. **col\_index\_num** = Which column in the lookup table has the items that you want to go and get and bring back to the cell? You have to count to determine which columns contains the items you want to retrieve: is it column 2, or column 3, or column 4, and so on.
- iv. **[range\_lookup]** = Because there are two different types of lookup, we must tell VLOOKUP which of the two lookups we want it to do: either: Approximate Match Lookup or Exact Match Lookup. This argument tells VLOOKUP how to find a match in the first column of the Lookup Table.

### 1. Approximate Match:

- For “**Approximate Match**” we must put = TRUE or 1 or omitted.




- How **Approximate Match** works:

1. For Approximate Match the VLOOKUP table MUST be sorted on the first column: Ascending, A to Z (Small to Big).
2. This is how Approximate Match Lookup works:
  - i. It will look through the first column:
    1. If the first value in the table is smaller than the lookup\_value, VLOOKUP returns a Not Available Error: #N/A!
    2. Then it looks through the first column until it bumps into the first value bigger than it and then jump back one row. When it finds a match, it knows what row it should look in.
      - i. It actually does a “binary search”, which is a technical computer term for “Approximate Match”. “Binary Search” calculates quickly compared to “Exact Match”.
    3. If the lookup\_value is bigger than the last value, it stops at the last row.

### 2. Exact Match:

- For “**Exact Match**” we must put = FALSE or 0.



- How **Exact Match** works:

1. VLOOKUP will look through each item in the first column of the VLOOKUP table and try to find a match. When it finds a match, it knows what row it should look in.
  2. If VLOOKUP cannot find a match it will give an #N/A! error that tells you it did not find a match “it is not available”.
- Note about Exact Match: If you have very large data sets, Exact Match Lookup may cause formula to calculate slowly because “Exact Match” Lookup must look through every item, one-by-one, until it finds a match.

## 12) Lookup Tables are Everywhere

1. Looking things up in Lookup Tables is a common task in business, accounting and other professions.
2. Almost all Lookup Tables are vertical because the items in the first column are listed vertically.
3. Examples of Looking up items in a Vertical Lookup Table:

What Region does "Miles, Josefina" represent?

Sales Rep	Region
Anderson, Robin	Midwest
Bryan, Viola	West
Gonzalez, Danny	East
Miles, Josefina	Southwest
Nguyen, Sheldon	East
Nichols, Claudia	Northwest
Pittman, Otis	West
Richardson, Kristina	Northwest
Rodgers, Bob	Midwest
Stokes, Taylor	Southwest

If your sales are \$6000, what is your commission?

Your Sales	Commission
\$0	\$0
\$5,000	\$200
\$12,500	\$625
\$15,000	\$975
\$20,000	\$1,700

If your Income is \$3000, what is your tax?

Income	Tax
\$0	\$0.00
\$1,000	\$25.00
\$2,000	\$60.00
\$5,000	\$120.00
\$10,000	\$175.00

What is the price for Quad?

Product	Supplier	Price
Flying Eagle	Channel Craft	\$19.95
V Range	Colorado	\$18.95
Quad	Gel Boom	\$43.95
Bellen	Gel Boom	\$26.95
Carlota	Gel Boom	\$27.95

Variable Piecework Lookup Table. Which category do we use for 120 items made?

Units Lower Limit	Units Upper Limit	Pay per Unit	Number Made Through Previous	Earnings Made Through Previous
0	50	\$1.00		
51	75	\$1.10	50	\$50.00
76	110	\$1.25	75	\$77.50
111	130	\$1.50	110	\$121.25
131	150	\$1.75	130	\$151.25
151	More	\$2.00	150	\$186.25

Variable Commission Rate Lookup Table. Which category do we use for sales of \$7,500?

Sales Lower Limit	Sales Upper Limit	% of Sales Earned	Amount Sales Through Previous Categories	Com Through Previous Categories
0	\$5,000.00	1.00%		
\$5,000.01	\$7,000.00	1.50%	\$5,000.00	\$50.00
\$7,000.01	\$9,000.00	2.50%	\$7,000.00	\$80.00
\$9,000.01	\$15,000.00	4.00%	\$9,000.00	\$130.00
\$15,000.01	\$25,000.00	5.00%	\$15,000.00	\$370.00
\$25,000.01	more	7.50%	\$25,000.00	\$870.00

What is "Chukes, Hal" Zip Code?

Employee	Address	City	State	Zip	E-mail
June, Aler	3557 1st St.	Seattle	WA	98114	JNV@yahoo.com
Acerio, Natisha	6281 173rd St.	Tacoma	WA	98131	HDT641@fun.edu
Bruess, Natisha	7149 1st Ave.	Seattle	WA	98133	YSE222@fun.edu
Chukes, Hal	7726 66th Ave.	Tacoma	WA	98111	MGF675@fun.edu
Dahnke, Georgeann	316 66th Blvd. 151st St.	Kent	WA	98124	RWN@fun.edu
Dillaman, Darius	965 151st St.	Kent	WA	98116	MJG@yahoo.com
Durtschi, Dane	7582 4th Lane	Tacoma	WA	98117	KVC@gmail.com
Fila, Bryon	1654 66th St.	Seattle	WA	98114	QKC@gmail.com
Fukumoto, Marvis	3653 4th Blvd.	Seattle	WA	98128	PKF@yahoo.com
General, Marlin	7900 173rd Lane	Kent	WA	98126	KFP@yahoo.com

### 13) Straight Commission Rates

	A	B	C	D	E	F	G	H
1	<b>Ex 13</b>	<b>Goal:</b> Calculate Gross Pay for Straight Commission Rate.						
2		<b>Contact reads:</b> Base monthly salary of \$3,000 plus 1.5% of the total sales they made for the month.						
3		Employee Name:	Sioux					
4		Base Pay	\$3,000.00					
5		Sioux's Sales for month:	\$25,000.00					
6		Straight Commission Rate	1.50%					
7		Gross Pay	\$3,375.00			=C4+C5*C6		

### 14) VLOOKUP & Exact Match Lookup for Straight Commission Rates When each Employee has Different Rate

	A	B	C	D	E	F	G	H	I
1	<b>Ex 14</b>	<b>Goal:</b> Calculate Gross Pay for Straight Commission Rate using VLOOKUP & Exact Match Lookup.							
2		<b>Contact reads:</b> Each employee has no base salary and has a Straight Commission Rate defined by their personal Employment Contract.							
3		Lookup Table below shows the commission rate for each employee.							
4									
5		<b>VLOOKUP Function Notes:</b>							
6		=VLOOKUP(lookup_value,table_array,col_index_num,[range_lookup])							
7		lookup_value: Item you are trying to find an <b>exact</b> match for in in first column of lookup table.							
8		table_array: Lookup Table (1st column has item to try and make an exact match with)							
9		col_index_num: Which Columns has the Number you want to go and get and bring back to cell? 1, 2?							
10		[range_lookup]: Exact Match (FALSE) allows VLOOKUP to find an exact match.							
11									
12					Formula is E16:	Formula is F16:			
13					=VLOOKUP(C16,\$H\$17:\$I\$26,2,FALSE)	=ROUND(D16*E16,2)			
14									
15		<b>Date</b>	<b>Employee</b>	<b>Sales</b>	<b>Commission Rate</b>	<b>Commission Amount</b>	<b>1</b>	<b>2</b>	
16		2/11/2018	Brandee Bussey	\$8,475.33	0.032	271.21	Employee	Commission Rate	
17		2/11/2018	Annis Dorris	\$7,813.40	0.014	109.39	Brandee Bussey	0.032	
18		2/11/2018	Noreen Vogel	\$9,924.32	0.017	168.71	Annis Dorris	0.014	
19		2/11/2018	Letty Bautista	\$11,749.65	0.016	187.99	Noreen Vogel	0.017	
20		2/11/2018	Jaclyn Mcfall	\$8,996.44	0.013	116.95	Letty Bautista	0.016	
21		2/11/2018	Missy Bauer	\$11,466.98	0.028	321.08	Jaclyn Mcfall	0.013	
22		2/11/2018	Kiesha Huggins	\$9,461.12	0.015	141.92	Missy Bauer	0.028	
23		2/11/2018	Lucinda Stitt	\$14,438.89	0.013	187.71	Kiesha Huggins	0.015	
24		2/11/2018	Brinda Hannon	\$8,290.61	0.035	290.17	Lucinda Stitt	0.013	
25		2/11/2018	Deandrea Ralph	\$11,235.01	0.013	146.06	Brinda Hannon	0.035	
26		2/12/2018	Brandee Bussey	\$13,915.89	0.032	445.31	Deandrea Ralph	0.013	
27		2/12/2018	Annis Dorris	\$11,397.23	0.014	159.56			
28		2/12/2018	Noreen Vogel	\$10,564.96	0.017	179.6			
44		2/13/2018	Brinda Hannon	\$5,897.22	0.035	206.4			
45		2/13/2018	Deandrea Ralph	\$11,200.01	0.013	145.6			
46									
47					<b>Total</b>	\$6,284.20			

### 15)VLOOKUP & Approximate Match Lookup for Variable Commission Rates

	A	B	C	D	E	F	G	H	I	
1	Ex 15	Goal: Build Variable Commission Rate Lookup Table. Calculate Gross Pay for Variable Commission Rate using VLOOKUP & Approximate Match Lookup.								
2		Contact reads: Each employee has no base salary and uses these Variable Commission Rates:								
3		Sales from \$0.00 - \$50,000.00 = 3.50%								
4		Sales from \$50,000.01 - \$125,000.00 = 4.00%								
5		Sales from \$125,000.01 - \$150,000.00 = 5.00%								
6		Sales from \$150,000.01 - \$175,000.00 = 6.50%								
7		Sales from \$175,000.01 - \$225,000.00 = 7.50%								
8		Sales from \$225,000.01 - more = 10.00%								
9										
10										
11		Formula is B17:			Formula is E17:		Formula is F17:			
12		=C16+0.01			=C16		=F16+ROUND((E17-E16)*D16,2)			
13										
14		1	2	3	4	5				
15		Sales Lower Limit	Sales Upper Limit	Commission Rate	Amount Sales Through Previous Categories	Com Through Previous Categories				
16		0	\$50,000.00	3.50%						
17		\$50,000.01	\$125,000.00	4.00%	\$50,000.00	\$1,750.00				
18		\$125,000.01	\$150,000.00	5.00%	\$125,000.00	\$4,750.00				
19		\$150,000.01	\$175,000.00	6.50%	\$150,000.00	\$6,000.00				
20		\$175,000.01	\$225,000.00	7.50%	\$175,000.00	\$7,625.00				
21		\$225,000.01	more	10.00%	\$225,000.00	\$11,375.00				
22										
23				Formula is D27:		Formula is E27:		Formula is F27:		
24				=VLOOKUP(C27,\$B\$16:\$F\$21,5)		=VLOOKUP(C27,\$B\$16:\$F\$21,4)		=VLOOKUP(C27,\$B\$16:\$F\$21,3)		
25								Formula is I27:		
26		Employee	Sales	Com Through Previous Categories	Amount Sales Through Previous Categories	Commission Rate	Gross Pay	Gross Pay		
27		Abdi Smitty	\$214,350.82	7625	175000	0.075	\$10,576.31	\$10,576.31		
28		Gigi Thompson	\$111,707.40	1750	50000	0.04	\$4,218.30	\$4,218.30		
29		Kenya Panther	\$182,654.10	7625	175000	0.075	\$8,199.06	\$8,199.06		
30		Dylan Minger	\$129,816.63	4750	125000	0.05	\$4,990.83	\$4,990.83		
31		Chin Yu	\$103,388.80	1750	50000	0.04	\$3,885.55	\$3,885.55		
32		Billy Smith	\$195,569.06	7625	175000	0.075	\$9,167.68	\$9,167.68		
33		Lin Pham	\$141,770.73	4750	125000	0.05	\$5,588.54	\$5,588.54		
34		Shelia Downings	\$184,464.47	7625	175000	0.075	\$8,334.84	\$8,334.84		
35										
36							Total	\$54,961.11		
37										



## 16)VLOOKUP & Approximate Match Lookup for Commission Rate Based on Amount of Sales Made by the Employee

- Sometimes an employee contract will pay an employee a Straight Commission Rate for all their sales, but the Rate will go up depending on the amount of sales they make.
  - The rate is not a Variable Rate where we have a different rate for each category, the employee will simply get paid the same Commission Rate for all the sales they make, but the single rate is determined by the amount of their sales.
- Example:

	A	B	C	D	E	F																
1	<b>Ex 16</b>	<b>Goal:</b> Calculate Gross Pay using VLOOKUP (Approximate Match)																				
2		for Commission Rate Based on Amount of Sales Made by Employee.																				
3		<b>Contract reads:</b> Sales person gets a base monthly salary of \$2,000 plus a percentage of the total sales																				
4		they made for the month based on the Commission Rate determined by the below table:																				
5		<b>Note:</b>	<table border="1"> <tr> <td>0.00 - \$4,999.99</td> <td>==&gt;&gt;</td> <td>0.50%</td> </tr> <tr> <td>\$5,000.00 - \$19,999.99</td> <td>==&gt;&gt;</td> <td>2.00%</td> </tr> <tr> <td>\$20,000.00 - \$29,999.99</td> <td>==&gt;&gt;</td> <td>2.50%</td> </tr> <tr> <td>\$30,000.00 - \$39,999.99</td> <td>==&gt;&gt;</td> <td>3.75%</td> </tr> <tr> <td>\$40,000 or more</td> <td>==&gt;&gt;</td> <td>5.25%</td> </tr> </table>			0.00 - \$4,999.99	==>>	0.50%	\$5,000.00 - \$19,999.99	==>>	2.00%	\$20,000.00 - \$29,999.99	==>>	2.50%	\$30,000.00 - \$39,999.99	==>>	3.75%	\$40,000 or more	==>>	5.25%		
0.00 - \$4,999.99	==>>	0.50%																				
\$5,000.00 - \$19,999.99	==>>	2.00%																				
\$20,000.00 - \$29,999.99	==>>	2.50%																				
\$30,000.00 - \$39,999.99	==>>	3.75%																				
\$40,000 or more	==>>	5.25%																				
6		The rate is not a Variable Rate where we have a																				
7		different rate for each category, the employee will																				
8		simply get paid the same Commission Rate for all the																				
9		sales they make, but the single rate is determined by the																				
10		amount of their sales.																				
11																						
12		<b>Employee Name:</b>	Kenya Freeman																			
13		<b>Base Pay</b>	\$2,000.00																			
14		<b>Kenya Freeman's Sales for month:</b>	\$143,958.25																			
15		<b>Gross Pay</b>	\$9,557.81																			
16					=VLOOKUP(C14,B18:C22,2)*C14+C13																	
17		<b>Lower Sales Limit</b>	<b>Commission Rate</b>																			
18		0	0.50%																			
19		5000	2.00%																			
20		20000	2.50%																			
21		30000	3.75%																			
22		40000	5.25%																			
23																						